

ΥΠΟΛΟΓΙΣΤΕΣ Ι

ΕΠΑΝΑΛΗΨΕΙΣ ΕΝΤΟΛΗ DO

Τι χρειάζεται η εντολή DO ;

Όταν απαιτείται να εκτελεστεί **πολλές φορές** το **ίδιο τμήμα** ενός προγράμματος.

Τετριμμένο παράδειγμα:

Κατασκευάστε πρόγραμμα που θα εμφανίζει στην οθόνη **10 φορές** τη λέξη HELLO.

Τι χρειάζεται η εντολή DO ;

```
PROGRAM HELLO  
  
WRITE (*,*) 'HELLO'  
WRITE (*,*) 'HELLO'  
WRITE (*,*) 'HELLO'  
WRITE (*,*) 'HELLO'  
WRITE (*,*) 'HELLO'  
WRITE (*,*) 'HELLO'  
WRITE (*,*) 'HELLO'  
WRITE (*,*) 'HELLO'  
WRITE (*,*) 'HELLO'  
WRITE (*,*) 'HELLO'  
  
END
```

Αν όμως έπρεπε το HELLO να εμφανιστεί **1000** φορές ;

Τι χρειάζεται η εντολή DO ;

```
PROGRAM HELLO2  
  IMPLICIT NONE  
  INTEGER K  
  
  DO K=1,10  
    WRITE (*,*) 'HELLO'  
  END DO  
  
END
```

Επανάλαβε τις εντολές μεταξύ DO και END DO 10 φορές.

Πως λειτουργεί η εντολή DO

```
DO K=1,10
    WRITE (*,*) 'HELLO'
END DO
```

Σε κάθε επανάληψη η μεταβλητή έχει διαφορετική τιμή.

- Στην εντολή DO υπάρχει μια ακέραια μεταβλητή (K), μια αρχική τιμή (1) και μια τελική τιμή (10).
- Μόλις το πρόγραμμα φτάσει στην εντολή DO, από την αρχική και τελική τιμή υπολογίζεται πόσες επαναλήψεις θα γίνουν (10 επαναλήψεις).
- Κατόπιν η μεταβλητή λαμβάνει την αρχική τιμή ($K=1$) και εκτελούνται για πρώτη φορά οι εντολές μέχρι το END DO.
- Το πρόγραμμα επιστρέφει πίσω στην εντολή DO.
- Η μεταβλητή αυξάνεται ($K=2$) και η διαδικασία επαναλαμβάνεται.

Παράδειγμα #1

Κατασκευάστε πρόγραμμα το οποίο θα εμφανίζει στην οθόνη την 2^η και 3^η δύναμη των ακεραίων από 1 έως 100.

Παράδειγμα #1

```
PROGRAM REPEAT
    IMPLICIT NONE
    INTEGER K

    DO K=1,100
        WRITE (*,*) K, K**2, K**3
    END DO

END
```

Στην εντολή WRITE μπορούμε να γράψουμε όχι μόνο μηνύματα και μεταβλητές, αλλά και ολόκληρες αριθμητικές παραστάσεις.

Συντακτικό της εντολής DO (1/2)

```
DO μεταβλητή = αρχική τιμή, τελική τιμή, βήμα
    εντολή1
    εντολή2
    ⋮
END DO
```

Σημείωση: κάθε εντολή γράφεται μερικά κενά (ή ένα tab) πιο δεξιά για ευκρίνεια.

Συντακτικό της εντολής DO (2/2)

Ορισμένες παρατηρήσεις για την εντολή DO:

- Η **μεταβλητή** πρέπει να είναι ακέραια.
- Το **βήμα** μπορεί να παραληφθεί, οπότε θεωρείται 1.
- Το **βήμα** μπορεί να λάβει και αρνητικές τιμές, όχι όμως 0.
- Η **μεταβλητή** αυξάνεται αυτόματα από την εντολή DO με το **βήμα** σε κάθε επανάληψη. Δεν επιτρέπεται η αλλαγή της μεταβλητής από εμάς (πχ. να γράψουμε $K=K+1$).
- Ανάλογα με την **αρχική τιμή**, **τελική τιμή** και **βήμα** μπορεί να μην γίνουν καθόλου επαναλήψεις.
- Εντός της εντολής DO επιτρέπεται να υπάρχουν άλλες εντολές DO.

9

Παράδειγμα #2

Πόσες επαναλήψεις θα γίνουν από τις παρακάτω εντολές DO ; Ποια είναι η τιμή της μεταβλητής σε κάθε επανάληψη ;

DO K=2, 9, 2 4 επαναλήψεις K=2,4,6,8

DO M=0, -5, -3 2 επαναλήψεις M=0,-3

DO N=15, 8, 2 Καμία επανάληψη

10

Παράδειγμα #3

Πόσες επαναλήψεις θα γίνουν από τις παρακάτω εντολές DO ; Ποια είναι η τιμή της μεταβλητής σε κάθε επανάληψη ;

- | | |
|---------------------|---------------------|
| 1) DO L=1, 7, 2 | 6) DO M=7, 4, -1 |
| 2) DO L=10, 17, 3 | 7) DO K=7, 1, 4 |
| 3) DO K=-2, 2 | 8) DO M=25, 20 |
| 4) DO N=1, 10, 25 | 9) DO J=-5, -10 |
| 5) DO K=-8, -10, -2 | 10) DO J=12, 12, -1 |

11

Παράδειγμα #3

- 1) DO L=1, 7, 2 4 επαναλήψεις L=1, 3, 5, 7
- 2) DO L=10, 17, 3 3 επαναλήψεις L=10, 13, 16
- 3) DO K=-2, 2 5 επαναλήψεις K=-2, -1, 0, 1, 2
- 4) DO N=1, 10, 25 1 επανάληψη N=1
- 5) DO K=-8, -10, -2 2 επαναλήψεις K=-8, -10

12

Παράδειγμα #3

- 6) DO M=7,4,-1 4 επαναλήψεις M=7,6,5,4
- 7) DO K=7,1,4 0 επαναλήψεις
- 8) DO M=25,20 0 επαναλήψεις
- 9) DO J=-5,-10 0 επαναλήψεις
- 10) DO J=12,12,-1 1 επανάληψη J=12

13

Παράδειγμα #4

Κατασκευάστε πρόγραμμα που θα τυπώνει τους τριγωνομετρικούς αριθμούς **ημίτονο**, **συνημίτονο** και **εφαπτομένη** για μια περιοχή γωνιών από M1 έως M2 μοίρες ανά μία μοίρα.
Τα M1, M2 θα εισάγονται από το πληκτρολόγιο.

Π.χ. για M1=0 και M2=5 το αποτέλεσμα θα πρέπει να είναι:

Γωνία	Ημίτονο	Συνημίτονο	Εφαπτομένη
0	0.00000000	1.00000000	0.00000000
1	0.01745241	0.99984770	0.01745506
2	0.03489950	0.99939083	0.03492077
3	0.05233596	0.99862953	0.05240778
4	0.06975647	0.99756405	0.06992681
5	0.08715574	0.99619470	0.08748866

14

Παράδειγμα #4

Υπενθύμιση:

Οι τριγωνομετρικές συναρτήσεις SIN(X), COS(X) και TAN(X) δέχονται το όρισμά τους σε ακτίνια.

Η μετατροπή από μοίρες σε ακτίνια γίνεται ως εξής:

$$A = \frac{\pi}{180} M$$

15

Παράδειγμα #4

```
PROGRAM TRIGON
  IMPLICIT NONE
  INTEGER M1, M2, M
  DOUBLE PRECISION PI, R
  DOUBLE PRECISION S, C, T
  WRITE (*,*) 'Εισάγετε τα M1, M2'
  READ (*,*) M1, M2
  PI = ACOS(-1.0D0)
  WRITE (*,*) 'Γωνία   Ημίτονο   Συνημίτονο   Εφαπτομένη'
  DO M = M1, M2
    R = M*PI/180
    S = SIN(R)
    C = COS(R)
    T = TAN(R)
    WRITE (*,*) M, S, C, T
  END DO
END
```

Παράδειγμα #4

Όμως το αποτέλεσμα θα είναι:

Γωνία	Ημίτονο	Συνημίτονο	Εφαπτομένη
0	0.	1.	0.
1	0.0174524064	0.999847695	0.0174550649
2	0.0348994967	0.999390827	0.0349207695
3	0.0523359562	0.998629535	0.0524077793
4	0.0697564737	0.99756405	0.0699268119
5	0.0871557427	0.996194698	0.0874886635



Ο πίνακας δεν είναι στοιχισμένος

17

Παράδειγμα #4

Αντικαθιστούμε την εντολή:

```
WRITE (*,*) M, S, C, T
```

με την εντολή :

```
WRITE (*, '(I6,3X,F11.8,3X,F11.8,3X,F11.8)') M, S, C, T
```


Πεδία μορφοποίησης
Για κάθε μεταβλητή υπάρχει ένα πεδίο μορφοποίησης που καθορίζει πως θα εμφανιστεί η μεταβλητή στην οθόνη.

18

Πεδία μορφοποίησης

I6

Ακέραιος αριθμός σε 6 θέσεις.




3X

Τρεις κενές θέσεις.

F11.8

Πραγματικός αριθμός σε 11 θέσεις εκ των οποίων 8 είναι δεκαδικά ψηφία.



19

Αθροίσματα

Μια σημαντική εφαρμογή της εντολής DO είναι ο υπολογισμός αθροισμάτων. Π.χ:

Για γνωστό N, να υπολογιστεί το άθροισμα:
 $S = 1+2+3+ \dots +N$

Υπενθύμιση:

Το άθροισμα δίνεται από τη σχέση: $S = \frac{N(N+1)}{2}$

20

Αθροίσματα

Χρησιμοποιούμε μια μεταβλητή S (ονομάζεται **αθροιστής**) στην οποία θα αποθηκεύσουμε το άθροισμα.

```
S = 0           ! Αρχική τιμή.
DO K=1, N
    S = S+K     ! Προσθήκη στο άθροισμα.
END DO
```

21

Αθροίσματα

Ποια είναι η τιμή του αθροιστή S στο τέλος κάθε επανάληψης ;

```
S = 0
DO K=1, N
    S = S+K
END DO
```

K	S
1	1
2	1+2
3	1+2+3
4	1+2+3+4
5	1+2+3+4+5
...	...
N	1+2+3+4+...+N

22

Παράδειγμα #5

Τροποποιήστε τον προηγούμενο κώδικα ώστε να υπολογίζει επιπλέον και τα αθροίσματα:

$$S2 = 1^2 + 2^2 + 3^2 + \dots + N^2$$

$$S3 = 1^3 + 2^3 + 3^3 + \dots + N^3$$

$$S4 = 1^4 + 2^4 + 3^4 + \dots + N^4$$

Σημείωση:

$$S2 = \frac{N(N+1)(2N+1)}{6}$$

$$S3 = \frac{N^2(N+1)^2}{4}$$

$$S4 = \frac{N(N+1)(2N+1)(3N^2+3N-1)}{30}$$

23

Παράδειγμα #5

```
S = 0
S2 = 0
S3 = 0
S4 = 0
DO K=1, N
    S = S + K
    S2 = S2 + K**2
    S3 = S3 + K**3
    S4 = S4 + K**4
END DO
```

} Αρχικές τιμές σε 4 αθροιστές.

24

Παράδειγμα #6

Γράψτε απόσπασμα προγράμματος για να υπολογίσετε το άθροισμα:

$$\sum_{k=1}^N \frac{2}{1+k^2}$$

Υπενθύμιση:

$$\sum_{k=1}^N \frac{2}{1+k^2} = \frac{2}{1+1^2} + \frac{2}{1+2^2} + \frac{2}{1+3^2} + \dots + \frac{2}{1+N^2}$$

25

Παράδειγμα #6

Προσοχή:

- Αριθμητής και παρονομαστής είναι **ακέραιοι**
- Το συνολικό άθροισμα είναι **πραγματικός**

```
A = 0.
DO K=1,N
    A = A + 2./(1+K**2)
END DO
```

26

Παράδειγμα #7

Γράψτε απόσπασμα προγράμματος για να υπολογίσετε το **γινόμενο**:

$$1 \cdot 2 \cdot 3 \cdot 4 \cdot \dots \cdot N$$

Υπενθύμιση:

Το γινόμενο αυτό ονομάζεται παραγοντικό και συμβολίζεται με **N!**

```
P = 1
DO K=2,N
    P = P*K
END DO
```

27

Παράδειγμα #8

Υπολογίστε το άθροισμα

$$\sum_{k=1}^N \frac{1}{k}$$

μόνο για τις περιττές τιμές k.

Σημείωση:

Το ζητούμενο άθροισμα είναι:

$$\frac{1}{1} + \frac{1}{3} + \frac{1}{5} + \dots + \frac{1}{N}$$

28

Παράδειγμα #8

```
S = 0.  
DO K=1,N,2  
    S = S+1./K  
END DO
```

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

29

Παράδειγμα #9

Υπολογίστε το άθροισμα:

$$\sum_{k=1}^N q_k \quad \text{όπου} \quad q_k = \begin{cases} k^2 & \text{αν } k \text{ άρτιος} \\ N^k & \text{αν } k \text{ περιττός} \end{cases}$$

Παρατήρηση:

Το ζητούμενο άθροισμα είναι:

$$\sum_{k=1}^N q_k = q_1 + q_2 + q_3 + \dots + q_N$$

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

30

Παράδειγμα #9

```
S = 0  
DO K=1,N  
    IF (MOD(K,2) == 0) THEN  
        QK = K**2  
    ELSE  
        QK = N**K  
    END IF  
    S = S + QK  
END DO
```

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

31

Παράδειγμα #9 (2^{ος} τρόπος)

Χωρίζουμε το άθροισμα σε δύο επιμέρους αθροίσματα:

$$S_1 = q_1 + q_3 + q_5 + \dots \quad \text{Άθροισμα περιπτών όρων (} q_k = N^k \text{)}$$

$$S_2 = q_2 + q_4 + q_6 + \dots \quad \text{Άθροισμα άρτιων όρων (} q_k = k^2 \text{)}$$

```
S1 = 0  
DO K=1,N,2  
    S1 = S1 + N**K  
END DO  
S2 = 0  
DO K=2,N,2  
    S2 = S2 + K**2  
END DO  
S = S1+S2
```

Άθροισμα περιπτών όρων

Άθροισμα άρτιων όρων

Ολικό άθροισμα

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

32

Παράδειγμα #10

Υπολογίστε το άθροισμα:

$$1 - \frac{1}{3} + \frac{1}{5} - \frac{1}{7} + \dots - \frac{1}{N}$$

Σημείωση:

Το άθροισμα συγκλίνει στο $\pi/4$.

Παρατηρήσεις:

- Οι παρανομαστές αυξάνουν κατά 2.
- Κάθε όρος έχει αντίθετο πρόσημο από τον προηγούμενο.

33

Παράδειγμα #10

Χωρίζουμε το άθροισμα σε δύο επιμέρους αθροίσματα:

$$S_1 = 1 + \frac{1}{5} + \frac{1}{9} + \frac{1}{13} + \dots \quad S_2 = -\frac{1}{3} - \frac{1}{7} - \frac{1}{11} - \frac{1}{15} \dots$$

```

S1 = 0
DO K=1,N,4
    S1 = S1 + 1./K
END DO
} Άθροισμα θετικών όρων

S2 = 0
DO K=3,N,4
    S2 = S2 - 1./K
END DO
} Άθροισμα αρνητικών όρων

S = S1 + S2
} Ολικό άθροισμα

```

34

Παράδειγμα #10 (2^{ος} τρόπος)

Εκτός από τον αθροιστή, ορίζουμε μια μεταβλητή για το πρόσημο η οποία παίρνει τις τιμές 1 ή -1.

```

S = 0.           ! Αθροιστής
P = 1.           ! Πρόσημο
DO K=1,N,2
    S = S+P/K     ! Άθροισμα
    P = -P        ! Αλλαγή προσήμου
END DO
WRITE (*,*) S    ! Εμφάνιση αθροίσματος

```

35

Αναδρομικές σχέσεις

Σειρές αριθμών όπου **ο κάθε όρος εξαρτάται από προηγούμενους όρους** περιγράφονται από αναδρομικές σχέσεις. Π.χ.

Δίνεται η σειρά αριθμών a_1, a_2, a_3, \dots
 με $a_1 = 1$ και $a_n = 1 + a_{n-1}$
 Ποιοι είναι οι επόμενοι όροι ;

$$a_2 = 1 + a_1 = 1 + 1 = 2$$

$$a_3 = 1 + a_2 = 1 + 2 = 3$$

$$a_4 = 1 + a_3 = 1 + 3 = 4$$

36

Αναδρομικές σχέσεις

Στο πρόγραμμα χρησιμοποιούμε μία μόνο μεταβλητή (και όχι μία για κάθε όρο). Η μεταβλητή αυτή περιέχει κάθε φορά τον τρέχοντα όρο της σειράς.

```

A = 1                ! Πρώτος όρος
DO K=2, N
    A = 1+A          ! Επόμενος όρος
    WRITE (*,*) A
END DO

```

37

Παράδειγμα #11

Κατασκευάστε πρόγραμμα που θα υπολογίζει το N-οστό όρο της ακολουθίας:

$$a_n = 2 + \frac{1}{a_{n-1}}$$

με πρώτο όρο $a_1 = 1$

Σημείωση:

Η ακολουθία αυτή συγκλίνει γρήγορα στην τιμή $1 + \sqrt{2}$

38

Παράδειγμα #11

```

A = 1                ! Πρώτος όρος
DO K=2, N
    A = 2+1./A        ! Επόμενος όρος
    WRITE (*,*) A
END DO

```

Για N=6 θα εμφανιστεί στην οθόνη:

```

3.0000000000000000
2.3333333333333333
2.42857142857143
2.41176470588235
2.41463414634146 ←  $\approx 1 + \sqrt{2}$ 

```

39

Παράδειγμα #12

Ένας άνθρωπος ξεκινά να περπατά με βήμα B εκατοστά. Μετά από κάθε βήμα αυξάνει το βήμα του κατά A% σε σχέση με το προηγούμενο βήμα. Εάν κάνει N βήματα πόση απόσταση θα έχει διανύσει συνολικά ;

40

Παράδειγμα #12

Ονομάζουμε τα διαδοχικά βήματα:

$$\beta_1, \beta_2, \beta_3, \dots, \beta_N$$

Τα βήματα μπορούν να εκφραστούν από την αναδρομική σχέση:

$$\beta_k = \beta_{k-1} + \frac{A}{100} \beta_{k-1} \quad \text{με γνωστό πρώτο όρο } \beta_1$$

Όμως στο πρόβλημα αυτό δεν ζητείται να βρεθούν οι όροι της ακολουθίας, αλλά το άθροισμα όλων των όρων, δηλαδή: $S = \beta_1 + \beta_2 + \beta_3 + \dots + \beta_N$

Παράδειγμα #12

```

PROGRAM STEP
  IMPLICIT NONE
  INTEGER K, N
  DOUBLE PRECISION B, A, S

  WRITE (*,*) 'Εισάγετε τα B, A, N'
  READ (*,*) B, A, N

  S = B                                     ! Αρχική τιμή αθροίσματος
  DO K=2,N
    B = B + A/100*B                         ! Νέο βήμα
    S = S + B                               ! Άθροιση βήματος
  END DO
  WRITE (*,*) 'Η συνολική απόσταση είναι',S
END
    
```

Παράδειγμα #13

Ακολουθία Fibonacci

Υπολογίστε το N-στο όρο της ακολουθίας:


$$F_n = F_{n-1} + F_{n-2}$$

με $F_0 = 0$ και $F_1 = 1$

Χρησιμοποιούμε δύο μεταβλητές:

- P για τον προηγούμενο όρο (F_{n-1})
- PP για τον προ-προηγούμενο όρο (F_{n-2})

Παράδειγμα #13



- Κάθε φορά:
- στη θέση PP αποθηκεύεται η P
 - στη θέση P αποθηκεύεται η A

```

PP = P
P = A
    
```

Παράδειγμα #13

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

```
PROGRAM FIBO
  IMPLICIT NONE
  INTEGER PP, P, A, N, I
  WRITE (*,*) 'Μέχρι ποιον όρο;'
  READ (*,*) N
  PP = 0
  P = 1
  WRITE (*,*) PP
  WRITE (*,*) P
  DO I=2,N
    A = PP+P
    WRITE (*,*) A
    PP = P
    P = A
  END DO
END
```

45

Παράδειγμα #14

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

Καταθέτουμε σε μια τράπεζα ένα αρχικό κεφάλαιο Κ. Το ποσό αυτό ανατοκίζεται στο τέλος κάθε μήνα με επιτόκιο Ε%. Κάθε μήνα μετά τον ανατοκισμό ακολουθεί ανάληψη ποσού Α αν υπάρχει επαρκές υπόλοιπο. Πόσο είναι το υπόλοιπο του λογαριασμού μετά από Ν μήνες ;

46

Παράδειγμα #14

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

```
PROGRAM BANK
  IMPLICIT NONE
  DOUBLE PRECISION K, E, A, T
  INTEGER N, I

  WRITE (*,*) 'Εισάγετε τα Κ, Ε, Α, Ν'
  READ (*,*) K, E, A, N

  DO I=1,N
    T = K*E/100           ! Μηνιαίος τόκος
    K = K+T               ! Προσθήκη στο κεφάλαιο
    IF (K >= A) K = K-A  ! Ανάληψη
  END DO
  WRITE (*,*) 'Το τελικό κεφάλαιο είναι', K
END
```

47

Παράδειγμα #15

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

Υπολογισμός ολοκληρώματος

Υπολογίστε προσεγγιστικά το ολοκλήρωμα


$$I = \int_a^b f(x) dx$$

όταν δίνεται η συνάρτηση $f(x)$ και τα όρια ολοκλήρωσης a, b χρησιμοποιώντας τον **κανόνα του παραλληλογράμμου**.

Το ζητούμενο ολοκλήρωμα ισούται με το εμβαδόν της περιοχής μεταξύ της συνάρτησης και του άξονα x .

48

Παράδειγμα #15



Αλγόριθμος:

1. Χωρίζουμε το διάστημα $[a, b]$ σε N ίσα υποδιαστήματα.
2. Υπολογίζουμε το εμβαδόν του κάθε ορθογώνιου παραλληλογράμμου θεωρώντας ως ύψος την τιμή της συνάρτησης στο μέσον του.
3. Αθροίζουμε όλα τα επιμέρους εμβαδά.

49

Παράδειγμα #15

Ποιο είναι το μήκος κάθε υποδιαστήματος ;


$$h = \frac{b-a}{N}$$

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

50

Παράδειγμα #15

Ποιο είναι το μέσον του k υποδιαστήματος ;



Το 1 ^ο υποδιάστημα ξεκινάει στο:	a
Το 2 ^ο υποδιάστημα ξεκινάει στο:	$a+h$
Το 3 ^ο υποδιάστημα ξεκινάει στο:	$a+2h$
Το 4 ^ο υποδιάστημα ξεκινάει στο:	$a+3h$
...	
Το k υποδιάστημα ξεκινάει στο:	$a+(k-1)h$

Το μέσον του k υποδιαστήματος είναι:

$$a+(k-1)h + \frac{h}{2} = a+kh - \frac{h}{2}$$

51

Παράδειγμα #15

Ποιο είναι το εμβαδόν του υποδιαστήματος k ;

$$E_k = h f\left(a+kh - \frac{h}{2}\right)$$

Ποιο είναι το συνολικό εμβαδόν ;

$$\sum_{k=1}^N E_k = \sum_{k=1}^N h f\left(a+kh - \frac{h}{2}\right)$$

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

52

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ DO

Παράδειγμα #15

```
PROGRAM INTEG
  IMPLICIT NONE
  DOUBLE PRECISION A, B, H, S, X, F
  INTEGER N, K
  WRITE (*,*) 'Εισάγετε τα A,B,N'
  READ (*,*) A, B, N
  H = (B-A)/N           ! Μήκος υποδιαστήματος
  S = 0.
  DO K=1,N
    X = A+K*H-H/2       ! Μέσον του υποδιαστήματος
    F = SIN(X)          ! Τιμή της συνάρτησης
    S = S+H*F           ! Άθροιση επιμέρους εμβαδού
  END DO
  WRITE (*,*) 'Ολοκλήρωμα = ', S
END
```

53

Παράδειγμα #16

Υπολογίστε το άθροισμα

$$1 + x + \frac{x^2}{2!} + \frac{x^3}{3!} + \frac{x^4}{4!} + \dots + \frac{x^N}{N!}$$

όταν δίνεται το x και το N.

Υπενθύμιση:

Το ανωτέρω άθροισμα είναι η σειρά Taylor του e^x

54

Παράδειγμα #16

Ο κάθε όρος του αθροίσματος είναι της μορφής:

$$\frac{x^k}{k!}$$

Ο αριθμητής υπολογίζεται ως:

$$A = X^{**}K$$

Ο παρανομαστής υπολογίζεται ως:

```
P = 1
DO M=2,K
  P = P*M
END DO
```

55

Παράδειγμα #16

```
PROGRAM EX
  IMPLICIT NONE
  DOUBLE PRECISION X, S, A
  INTEGER N, M, K, P
  WRITE (*,*) 'Εισάγετε τα X, N'
  READ (*,*) X, N
  S = 1.
  DO K=1,N
    A = X**K
    P = 1
    DO M=2,K
      P = P*M
    END DO
    S = S + A/P
  END DO
  WRITE (*,*) S
END
```

Εξωτερικό DO για κάθε όρο

Εσωτερικό DO για το k!

56

Παράδειγμα #16

Πόσες πράξεις γίνονται στο πρόγραμμα ;

```

DO K=1,N
  A = X**K      → 1 ύψωση σε δύναμη
  P = 1
  DO M=2,K
    P = P*M     } K-1 πολλαπλασιασμοί
  END DO
  S = S + A/P   → 1 πρόσθεση, 1 διαίρεση
END DO
    
```

Συνολικά εντός του εξωτερικού DO γίνονται **K+2 πράξεις**

Παράδειγμα #16

K	Πράξεις (K+2)
1	1+2
2	2+2
3	3+2
4	4+2
5	5+2
...	...
N	N+2


Σύνολο πράξεων: $\frac{N(N+1)}{2} + 2N = \frac{N^2 + 5N}{2}$

Παράδειγμα #16 (2ος τρόπος)

$$1 + x + \frac{x^2}{2!} + \frac{x^3}{3!} + \frac{x^4}{4!} + \dots + \frac{x^N}{N!}$$

Παρατηρείστε ότι ο κάθε όρος του αθροίσματος μπορεί να γραφεί σε σχέση με τον προηγούμενο.

Παράδειγμα:



Παράδειγμα #16 (2ος τρόπος)

Γενικά: $\frac{x^k}{k!} = \frac{x^{k-1}}{(k-1)!} \frac{x}{k}$

$\text{όρος}_k = \frac{x}{k} \text{όρος}_{k-1} \quad \text{όρος}_0 = 1$



Αναδρομική σχέση με ένα προηγούμενο όρο.

Προσοχή: δεν ζητούνται οι επιμέρους όροι αλλά το άθροισμά τους.

Παράδειγμα #17

Γενικά:

$$\text{όρος τάξης } k = -\frac{x^2}{k(k-1)} \cdot \text{όρος τάξης } k-2$$



Αναδρομική σχέση με ένα προηγούμενο όρο.

Προσοχή: δεν ζητούνται οι επιμέρους όροι αλλά το άθροισμά τους.

65

Παράδειγμα #17

```
PROGRAM COSINE
  IMPLICIT NONE
  DOUBLE PRECISION X, S, T
  INTEGER N, K
  WRITE (*,*) 'Εισάγετε τα X, N'
  READ (*,*) X, N
  S = 1.                                ! Αρχική τιμή
  T = 1.                                ! Πρώτος όρος
  DO K=2,N,2
    T = -T*X**2/(K*(K-1)) ! Επόμενος όρος
    S = S + T              ! Άθροιση
  END DO
  WRITE (*,*) S
END
```

66

Παράδειγμα #18

Υπολογίστε το άθροισμα

$$x - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!} + \dots - \frac{x^N}{N!}$$

όταν δίνεται το x και το N.

Υπενθύμιση:

Το ανωτέρω άθροισμα είναι η σειρά Taylor του $\sin x$

67

Παράδειγμα #18

$$x - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!} + \dots - \frac{x^N}{N!}$$

Κάθε όρος μπορεί να γραφεί σε σχέση με τον προηγούμενο:

$$\text{όρος τάξης } k = -\frac{x^2}{k(k-1)} \cdot \text{όρος τάξης } k-2$$

Όμως προσοχή: ο πρώτος όρος είναι x

68

Παράδειγμα #18

```
PROGRAM SINE
  IMPLICIT NONE
  DOUBLE PRECISION X, S, T
  INTEGER N, K
  WRITE (*,*) 'Εισάγετε τα X, N'
  READ (*,*) X, N
  S = X                                ! Αρχική τιμή
  T = X                                ! Πρώτος όρος
  DO K=3,N,2
    T = -T*X**2/(K*(K-1))              ! Επόμενος όρος
    S = S + T                          ! Άθροιση
  END DO
  WRITE (*,*) S
END
```